

Eine Absatzformatvorlage in ein Zieldokument importieren

Dr. Volker Thormählen, 14. Aug. 2020

Inhalt

| | | |
|-----|---------------------------------------|---|
| 1 | Aufgabe | 2 |
| 2 | Lösung | 2 |
| 2.1 | Prozedur 'ImportParagraphStyle' | 2 |
| 2.2 | Funktionsprozedur 'StyleExists' | 3 |
| 3 | Ergebnis..... | 4 |
| 4 | Erweiterung..... | 4 |
| 5 | Literaturverzeichnis..... | 5 |

Abbildungen

| | |
|---|---|
| Abbildung 1: Beschreibung der Absatzformatvorlage 'Quellcode' | 4 |
|---|---|

Listings

| | |
|--|---|
| Listing 1: Eine best. Absatzformatvorlage in ein Zieldokument importieren..... | 3 |
| Listing 2: Funktionsprozedur zur Existenzprüfung einer Formatvorlage | 4 |
| Listing 3: Alle Formatvorlagen in der Dokumentvorlage des Quelldokuments in das Zieldokument importieren | 4 |

1 Aufgabe

In diesem Beitrag wird beschrieben, wie eine Formatvorlage¹ (kurz: FV) mit einer VBA²-Prozedur³ in ein WORD-Zieldokument importiert werden kann. Die Existenz eines WORD-Quelldokuments, das die zu importierende FV enthält, wird zwar vorausgesetzt, aber trotzdem wird eine Existenzprüfung der FV vorgenommen.

Abschließend wird Gliederungspunkt „Erweiterung“ (s. Seite 4) beschreiben, wie *alle* Formatvorlagen in der Dokumentvorlage⁴ des Quelldokuments auf einmal in das Zieldokument importiert werden können.

2 Lösung

2.1 Prozedur 'ImportParagraphStyle'

In dieser Prozedur beschreiben die grau markierten Kommentare den jeweiligen Verarbeitungsschritt.

Die Lösung wird in der Prozedur 'ImportParagraphStyle' mithilfe der sog. 'Application.OrganizerCopy-Methode' erzielt (s. Listing 1, Codezeile 62 ff).

```
1 Option Explicit
2 Sub ImportParagraphStyle()
3 ' Aufgabe: Best. Absatzformatvorlage in ein Zieldokument importieren.
4 ' Konstanten
5   ' Name der aktuellen Prozedur
6   Const conProcNm As String = "ImportParagraphStyle"
7   ' Vollständiger Pfad zum Quelldokument
8   Const conSrcPath As String = "C:\Users\volker\Documents\Absatzformatvorlage.docx"
9   ' Vollständiger Pfad zum Zieldokument
10  Const conTgtPath As String = "C:\Users\volker\Documents\NeuesTestDoc.docm"
11  ' Name der zu importierenden Formatvorlage im Quelldokument
12  Const conStyleNm As String = "Quellcode"
13
14 ' Variablen
15   Dim docSrc As Document ' Quelldokument
16   Dim docTgt As Document ' Zieldokument
17   Dim lngResponse As Long ' Antwort d. Benutzers auf Abfrage mit MsgBox
18
19 ' Zieldokument öffnen
20   Set docTgt = Documents.Open(FileName:=conTgtPath, ReadOnly:=False)
21
22 ' Pfad zum Quelldokument auf Gültigkeit prüfen.
23   If Dir(conSrcPath) = vbNullString Then
24     MsgBox "Abbruch: Der vollst. Pfad '" & conSrcPath & _
25         "' zur Quelldatei ist ungültig", vbCritical, conProcNm
26     GoTo ExitPoint
27   End If
28
29 ' Quelldokument öffnen
30   Set docSrc = Documents.Open(FileName:=conSrcPath, ReadOnly:=True)
31 ' Prüfen, ob das Quelldokument <docSrc> die FV <conStyleNm> enthält.
32   If Not StyleExists(docSrc, conStyleNm) Then
```

¹ Eine *Formatvorlage* speichert Formatierungsmuster unter einem bestimmten Namen, um sie später wieder verwenden zu können.

² VBA steht für *Visual Basis for Applications*, eine proprietäre Skriptsprache von Microsoft für die sog. Office-Suite.

³ Als *Prozedur* wird eine benannte Folge von VBA-Anweisungen bezeichnet, die mit ihrem Namen aufgerufen und als Einheit ausgeführt wird.

⁴ Eine *Dokumentvorlage* speichert Dokumente, die in ähnlicher Form immer wieder verwendet werden können, unter einer bestimmten Dateiendung. Sie enthält u. a. eigene oder angepasste Formatvorlagen.

```

33     MsgBox "Abbruch: Das Quelldokument '" & docSrc.Name & _
34         "' enthält NICHT die zu kopierende FV '" & _
35         conStyleNm & "!", vbExclamation, conProcNm
36     GoTo ExitPoint
37 End If
38
39 ' Prüfen, ob das Zieldokument <docTgt> die FV <conStyleNm> bereits enthält.
40 If StyleExists(docTgt, conStyleNm) Then
41     MsgBox "Warnung: Das Zieldokument '" & docTgt.Name & _
42         "' enthält bereits die zu kopierende Formatvorlage '" & _
43         conStyleNm & "!", vbCritical, conProcNm
44     lngResponse = MsgBox("Die bereits vorhandene Formatvorlage '" & conStyleNm & _
45         "' aus dem Zieldokument entfernen?", vbQuestion + vbYesNo)
46     If lngResponse = vbYes Then
47         Err.Clear
48         On Error Resume Next
49         docTgt.Styles(conStyleNm).Delete
50         If Err.Number > 0 Then
51             MsgBox "Abbruch: Das Entfernen der Formatvorlage '" & conStyleNm & _
52                 "' aus dem Zieldokument '" & docTgt.Name & _
53                 "' ist fehlgeschlagen", vbCritical, conProcNm
54             GoTo ExitPoint
55         End If
56     Else
57         GoTo ExitPoint
58     End If
59 End If
60
61 ' Die Formatsvorlage <conStyleNm> kopieren
62 On Error Resume Next
63 Application.OrganizerCopy _
64     Source:=docSrc, _
65     Destination:=docTgt, _
66     Name:=conStyleNm, _
67     Object:=wdOrganizerObjectStyles
68
69 ' Alle Formatvorlagen in der Dokumentvorlage kopieren
70 ' Call CopyAllStyles(docSrc, docTgt)
71
72 ' Abschließende Meldung ausgeben
73 If Not StyleExists(docTgt, conStyleNm) Then
74     MsgBox "Das Kopieren der Formatvorlage '" & conStyleNm & "' nach '" & _
75         docTgt.Name & " ist fehlgeschlagen!", vbExclamation + vbCritical
76 Else
77     MsgBox "Die Prozedur '" & conProcNm & "' wurde erfolgreich ausgeführt!", _
78         vbExclamation, conProcNm
79     MsgBox docTgt.Styles(conStyleNm).Description, vbInformation, conProcNm
80 End If
81
82 ExitPoint:
83 On Error Resume Next
84 docSrc.Close SaveChanges:=wdDoNotSaveChanges
85 Set docSrc = Nothing
86 docTgt.Close SaveChanges:=wdSaveChanges
87 Set docTgt = Nothing
88 End Sub

```

Listing 1: Eine best. Absatzformatvorlage in ein Zieldokument importieren

2.2 Funktionsprozedur 'StyleExists'

Die Prozedur '*ImportParagraphStyle*' benötigt die Funktionsprozedur '*StyleExists*' (s. Listing 2). Diese wird dreimal aufgerufen (s. Codezeile 32, 40 bzw. 73). Sie dient zur *Existenzprüfung* einer definierten *Formatvorlage* in einem bestimmten WORD-Dokument:

```

1 Function StyleExists(objDoc As Word.Document, strStyleNm As String) As Boolean
2     ' Zurückgeben, ob die Formatvorlage <strStyleNm> im Dokument <objDoc> vorkommt.
3     StyleExists = False
4     On Error Resume Next
5     StyleExists = objDoc.Styles(strStyleNm).NameLocal = strStyleNm
6 End Function

```

Listing 2: Funktionsprozedur zur Existenzprüfung einer Formatvorlage

3 Ergebnis

Bei fehlerfreiem Ablauf der Prozedur 'ImportParagraphStyle' enthält das WORD-Zieldokument die aus dem WORD-Quelldokument importierte Formatvorlage namens 'Quellcode'. Codezeile 79 in Listing 1 (vgl. dort) bewirkt folgende Beschreibung der soeben genannten Formatvorlage:




Abbildung 1: Beschreibung der Absatzformatvorlage 'Quellcode'

Die Beschreibung der Basisformatvorlage 'Standard' (s. Abbildung 1) kann ggf. zusätzlich angezeigt werden.

4 Erweiterung

Wenn nicht nur eine, sondern alle Formatvorlagen in der Dokumentvorlage des Quelldokuments in das Zieldokument importiert werden sollen, kann folgende Prozedur (s. Listing 3) genutzt werden:

```

1 Sub CopyAllStyles(docSrc As Document, docTgt As Document)
2     ' Alle FV in der Dokumentvorlage des Quelldokuments in ein Zieldokument kopieren.
3     Dim strPath As String
4     strPath = docSrc.AttachedTemplate.FullName
5     docTgt.CopyStylesFromTemplate Template:=strPath
6 End Sub

```

Listing 3: Alle Formatvorlagen in der Dokumentvorlage des Quelldokuments in das Zieldokument importieren

Die entsprechende Anweisung in der aufrufenden Prozedur lautet dann:

```
Call CopyAllStyles(docSrc, docTgt)
```

5 Literaturverzeichnis

- [1] o. V., „Application.OrganizerCopy-Methode,“ 08 06 2017. [Online]. Available: <https://docs.microsoft.com/de-de/office/vba/api/word.application.organizercopy>. [Zugriff am 11 08 2020].
- [2] V. Chen, „How to Copy Styles from One Word Document to Another,“ [Online]. Available: <https://www.datanumen.com/blogs/copy-styles-one-word-document-another/>. [Zugriff am 09 08 2020].
- [3] Avantix Learning Team, „How to Copy Styles Between Microsoft Word Documents Using the Organizer,“ 20 07 2020. [Online]. Available: <http://www.avantixlearning.ca/courses/save-time-by-copying-styles-between-microsoft-word-documents-training/>. [Zugriff am 08 08 2020].